# $Id$
# Maintainer: Allan McRae <allan@archlinux.org>

pkgname=dejagnu
pkgver=1.6
pkgrel=2
pkgdesc="Framework for testing other programs"
arch=('any')
url="http://www.gnu.org/software/dejagnu/"
license=('GPL')
depends=('sh' 'expect')
source=(https://ftp.gnu.org/gnu/dejagnu/${pkgname}-${pkgver}.tar.gz{,.sig})
md5sums=('1fdc2eb0d592c4f89d82d24dfdf02f0b'
         'SKIP')
validpgpkeys=('CB0B31452BBE862933013D9176362888B55988D4')   # Ben Elliston

build() {
  cd $srcdir/$pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

check() {
  cd $srcdir/$pkgname-$pkgver
  make check
}

package() {
  cd $srcdir/$pkgname-$pkgver
  make DESTDIR=$pkgdir/ install
}
